

## Maximal credit

Given integer array  $A$  of length  $N$ ,  $0 \leq N$ , we are asked to compute the maximal credit of  $A$ . Formally, for natural numbers  $p$  and  $q$  and integer  $r$  our program must establish

$$R: \quad r = \langle \uparrow p, q: 0 \leq p \leq q \leq N: C.p.q \rangle$$

where, for natural  $i$ ,  $C$  is defined as

$$C.p.q = \langle \#i: p \leq i < q: A.i > 0 \rangle - \langle \#i: p \leq i < q: A.i < 0 \rangle$$

\* \* \*

We investigate our standard invariant,  $P$ , which is the conjunction of

$$Pr: \quad r = \langle \uparrow p, q: 0 \leq p \leq q \leq n: C.p.q \rangle$$

$$Pn: \quad 0 \leq n \leq N$$

which we initially establish with  $n, r := 0, 0$ . We are heading for a program of the form

```

n, r := 0, 0
; do n ≠ N →
    “Increase n by 1 under invariance of P”
od

```

The proofs of termination and maintenance of  $Pn$  under  $n := n + 1$  are standard and trivial and we thus omit them.

\* \* \*

We observe

```

[[ Context: P ∧ n ≠ N
   ⟨↑ p, q: 0 ≤ p ≤ q ≤ n + 1: C.p.q⟩
= { split off q = n + 1, requires n + 1 ≤ N which follows from Pn ∧ n ≠ N ; Pr }
   r ↑ ⟨↑ p: 0 ≤ p ≤ n + 1: C.p.(n + 1)⟩
= { assuming (n := n + 1).Ps }
   r ↑ s
]]

```

where we strengthen  $P$  with

$$Ps: \quad s = \langle \uparrow p: 0 \leq p \leq n: C.p.n \rangle$$

which we establish initially with  $n, s := 0, 0$ .

At this point we have

$$\begin{aligned} & \llbracket \text{Context: } P \wedge n \neq N \\ & \quad \{ Pr \wedge (n := n+1).Ps \} \quad r := r \uparrow s \quad \{ (n := n+1).Pr \} \\ & \rrbracket \end{aligned}$$

giving us a program of the form

```

n, r, := 0, 0
; do n ≠ N →
    “Establish (n := n+1).Ps ”
    ; n, r := n+1, r ↑ s
od

```

\* \* \*

We turn our attention to maintenance of  $Ps$ . We observe

$$\begin{aligned} & \llbracket \text{Context: } P \wedge n \neq N \\ & \quad \langle \uparrow p : 0 \leq p \leq n+1 : C.p.(n+1) \rangle \\ & = \quad \{ \text{split off } p = n+1, \text{ requires } 0 \leq n \wedge n+1 \leq N \text{ which follow from } Pn \} \\ & \quad \langle \uparrow p : 0 \leq p \leq n : C.p.(n+1) \rangle \uparrow C.(n+1).(n+1) \\ & = \quad \{ \text{properties of } C ; + \text{ over } \uparrow \} \\ & \quad \left\{ \begin{array}{ll} (\langle \uparrow p : 0 \leq p \leq n : C.p.n \rangle + 1) \uparrow 0, & \text{if } 0 < A.n \\ (\langle \uparrow p : 0 \leq p \leq n : C.p.n \rangle - 1) \uparrow 0, & \text{if } A.n < 0 \\ (\langle \uparrow p : 0 \leq p \leq n : C.p.n \rangle) \uparrow 0, & \text{if } A.n = 0 \end{array} \right. \\ & = \quad \{ Ps \} \\ & \quad \left\{ \begin{array}{ll} (s+1) \uparrow 0, & \text{if } 0 < A.n \\ (s-1) \uparrow 0, & \text{if } A.n < 0 \\ s \uparrow 0, & \text{if } A.n = 0 \end{array} \right. \\ & \rrbracket \end{aligned}$$

We now have

```
[[ Context: P ∧ n ≠ N
  { Ps }
  if 0 < A.n → s := (s + 1) ↑ 0
  [] A.n < 0 → s := (s - 1) ↑ 0
  [] A.n = 0 → s := s ↑ 0
  fi
  { (n := n + 1).Ps }
  ]]
```

Hence our solution

```
n, r, s := 0, 0, 0
do n ≠ N →
  if 0 < A.n → s := (s + 1) ↑ 0
  [] A.n < 0 → s := (s - 1) ↑ 0
  [] A.n = 0 → s := s ↑ 0
  fi
  ; n, r := n + 1, r ↑ s
od
```

*E. Emmanuel Macaulay*  
eric@mathmeth.com

*16th January 2007*